Van der Waals, Korteweg, van Laar: a Maple Excursion into the Thermodynamics of Binary Systems[1]

A.H.M. Levelt

University of Nijmegen
Mathematisch Instituut
Toernooiveld 1, 6525 ED Nijmegen
The Netherlands

Abstract

The computer algebra system MAPLE is used to redo some old and more recent computations in thermodynamics in a rather easy way.

1 Introduction

In 1890 J.D. van der Waals published a paper on the thermodynamic properties of liquid mixtures. He extended his famous equation of state to this new situation. In the sequel we restrict ouselves to a mixture of two pure substances (components) which may be in liquid or gaseous phase. A unary system (i.e. one component system) has one critical point. A binary mixture in general has a critical curve. Obviously, it was an important problem to compute this critical curve. Van der Waals found approximations. J.D. Korteweg has published several papers on the subject ([5], [6]), but has not given the equation for the critical curve. It was the 'mathematical chemist' J.J. van Laar who solved the problem in 1905 ([8]). It was a difficult computation as appears from a letter to Lorentz:

'You should see all the paper on which I made the calculations which failed!'

(cited from the interesting recent biography of van Laar, [2], page 80.) Shortly afterwards van Laar published a second paper ([9]) on the properties of the critical curve. He found that for mixtures of some substances the critical curve can have a singular point.

For someone involved in computer algebra van Laar's papers are a challenge: how to reproduce van Laar's results as efficiently as possible using a computer algebra system? This is more than a play or a sport. If successful one may try to use computer algebra system for solving actual problems in thermodynamics. In this paper I show how to find van Laar's results using the computer algebra system MAPLE. Van Laar made a special assumption (the geometric mean condition). More recently P.H. van Konynenburg and R.L. Scott (cf. [4]) dropped this assumption and were nevertheless able to find the equation of the critical curve. This a more complicated computation than van Laar's. I do not know how they made their calculations. Though they used the computer heavily for their numerical calculations, they do not mention symbolic calculation on a computer. They call the critical curve calculation 'tedious but straightforward algebra' and found their result after 'much lengthy algebra'. I was able to derive the equation of the critical curve (formula (A1) on page 533 of [4]) by the method of the present paper. For those interested a program is available.

Because of lack of space and time it was impossible to make this paper self-contained. A very brief explanation of relevant thermodynamical theory is given. For further theory cf. [1],[3],[7]. Lacking knowledge of thermodynamics is no obstacle to understand the computer algebra in this paper.

Computer algebra seems to be used more and more as a tool in thermodynamics, not only for pure formula manipulation, but also for the generation of efficient numerical code (cf. [11]).

[10] and [2] are references to the (early) history of thermodynamics and the important role Dutch scientists have played.

[1] Lecture at the seminar SCAFI, Amsterdam 5-6 November 1992. Corrected 21 January 1993

2 Thermodynamic theory and the basic equations

For the basic equations we shall follow van Laar's paper [8] with some changes of notation. In order to describe the thermodynamics of a mixture of two substances, coexisting in two phases (gaseous and liquid), 4 variables are needed: pressure (p), volume (V), temperature (T) and the molar fraction of the second substances (x) with respect to the total (molar) mass. These variables are not independent, but bound by an *equation of state*, which is in van Laar's paper the equation of van der Waals

$$R\,T = (p + \frac{a}{V^2})\,(V - b). \tag{1}$$

For a one component system a and b are positive constants, and the equation describes the gaseous phase, the liquid phase and also the situation of instable and metastable coexistence of the two phases. Additional theory on equilibria (*Maxwell's rule*) completes the description in case of coexistence. As is well-known van der Waals's equation only gives an approximative agreement with experiments. However as a qualitative description of real systems it is often useful.

The beautiful idea of van der Waals was to apply his equation also to the case of a binary mixture. For this he put

$$a = a_1(1-x)^2 + 2a_{12}x(1-x) + a_2x^2, \tag{2}$$

$$b = b_1(1-x) + b_2x. \tag{3}$$

This is not the place to discuss the merits of van der Waals' *one-fluid* appoach. We shall simply take it for granted. An additional assumption in van Laar's papers is the *geometric mean law*

$$a_1a_2 = a_{12}^2. \tag{4}$$

Sometimes another hypothesis is made: *equally sized molecules*, i.e.

$$b_1 = b_2. \tag{5}$$

A final piece in the description is some thermodynamical function: e.g. the Helmholtz or Gibbs free energy. Van Laar introduces a function ω of the variables p, T, x which is closely related to the Gibbs free energy (up to the sign). Van Laar's expression for ω in the one-fluid theory is

$$\omega = RT\log(V-b) + \frac{a}{v} - pV. \tag{6}$$

The dependence on x is hidden in a and b. From ω one can derive other important quantities like the *chemical potentials* of the substances. An expression for μ_1, the chemical potential of the first substance, is

$$\mu_1 = C(T) - \left(\omega - x\frac{\partial\omega}{\partial x}\right) + RT\log(1-x). \tag{7}$$

Here $C(T)$ is a function of T alone. μ_1 should be viewed as function of p, T, x. The first and second partial derivatives of μ_1 are the expressions van Laar was interested in. By elimination of p from the van der Waals equation and the equation $\partial\mu_1/\partial x = 0$ one finds a relation between V, T and x, which gives for fixed values of T the *spinodal curves* (in the x, V plane). Eliminating p, T from the van der Waals equation, $\partial\mu_1/\partial x = 0$ and $\partial^2\mu_1/\partial x^2 = 0$ one finds a relation between V, x, the (V, x projection of the) *critical curve* (*plaitpoint curve* in the ancient terminology). Let us express the partial derivatives of μ_1 in terms of ω using (7).

$$\frac{\partial\mu_1}{\partial x} = x\frac{\partial^2\omega}{\partial x^2} - \frac{RT}{1-x}, \tag{8}$$

$$\frac{\partial^2\mu_1}{\partial x^2} = x\frac{\partial^3\omega}{\partial x^3} + \frac{\partial^2\omega}{\partial x^2} - \frac{RT}{(1-x)^2}. \tag{9}$$

Hence we have

$$\frac{\partial\mu_1}{\partial x} = 0 \Longleftrightarrow x(1-x)\frac{\partial^2\omega}{\partial x^2} - RT = 0, \tag{10}$$

$$\frac{\partial^2 \mu_1}{\partial x^2} = 0 \Longleftrightarrow x(1-x)^2 \frac{\partial^3 \omega}{\partial x^3} + (1-x)^2 \frac{\partial^2 \omega}{\partial x^2} - RT = 0. \tag{11}$$

However,

$$\frac{\partial^2 \mu_1}{\partial x^2} = 0 \Longleftrightarrow x(1-x) \frac{\partial^3 \omega}{\partial x^3} + (1-2x) \frac{\partial^2 \omega}{\partial x^2} = 0 \tag{11'}$$

is more convenient in the computation.

Here follow the equations of van Laar. The spinodal (formula (6) on page 657 of [8]):

$$RT = \frac{2}{V^3} \left[x(1-x)\theta^2 + a(V-b)^2 \right], \tag{12}$$

where

$$\theta = b_1 \sqrt{a_2} - b_2 \sqrt{a_1} + \alpha(V-b), \tag{13}$$

$$\alpha = \sqrt{a_2} - \sqrt{a_1}. \tag{14}$$

Remark. There is a misprint in the original paper: the second α in van Laar's formula (6) on page 652 of loc. cit. should read a.

The equation for the critical curve (formula (8) on page 657)

$$\begin{aligned} crit \quad = \quad & x(1-x)\theta^3 \left[(1-2x)V - 3x(1-x)\beta \right] \\ & + \sqrt{a}(V-b)^2 \left[3x(1-x)\theta(\theta - \beta\sqrt{a}) + a(V-b)(V-3b) \right] = 0, \end{aligned} \tag{15}$$

where

$$\beta = b_2 - b_1. \tag{16}$$

In a second paper ([9]) van Laar investigates the possible shapes of the critical curve in the special case of equally sized molecules (cf. (5)). In particular he discovers that for some special value of a_1, a_2 the critical curve has a singular point, an ordinary double point. Again his computations are ingenious. He starts from the equation *crit* of the critical curve, makes some substitutions and finds a transformed equation f in new variables x, y. Then he solves the system of polynomial equations

$$f = 0, \frac{\partial f}{\partial x} = 0, \frac{\partial f}{\partial y} = 0,$$

and finds the relation that x must satisfy

$$x = \frac{1}{2} + \frac{1}{4} \left(\sqrt{2} - \sqrt{6} \right). \tag{17}$$

Then the ratio a_2/a_1 can be computed.

One may try to use the brute force of computer algebra to find the singular critical curve by solving directly the system of polynomial equations

$$crit = 0, \frac{\partial crit}{\partial V} = 0, \frac{\partial crit}{\partial x} = 0 \tag{18}$$

in the unknowns x, V, c_1, c_2. A clean technique is that of Gröbner bases. Using MAPLE one tries e.g.

```
> gbasis({crit,diff(crit,x),diff(crit,V)},{x,V,c1,c2},plex):
```

However, this does not work well. I have not been able to get a result along these lines. However, elimination using resultants leads rather quickly to a result.

3 Van Laar's results obtained using MAPLE

3.1 The equations of the spinodal and the critical curve

We shall show now how MAPLE can be used to derive van Laar's results with a minimum of effort.

Some introductory remarks. We shall follow as much as possible the notations of van Laar's papers. Due to the limited character set of MAPLE w is used instead of ω. The partial derivatives with respect to x are denoted $wx, wxx, \ldots$. In the MAPLE program $a1$ stands for a_1, etc. and al for α, be for β, ga for γ, th for θ. Since the temperature T only appears in the combination RT, we shall use RT as a MAPLE variable instead of T. The geometric mean law (4) is assumed, but (5) need not hold. Replacing a_1 by c_1^2 and a_2 by c_2^2 we can write $a = c^2$ where $c = (1-x)c_1 + xc_2$.

The MAPLE general differential operator D is used for the partial derivations with respect to x. This is done by first assigning $D(q) := 0$ for all quantities q not depending on x.

MAPLE has the very useful procedure 'collect'. An example. Let f a polynomial in x, y, z, w. Then

```
collect(f,[x,y],distributed,factor);
```

returns a polynomial in x, y in distributed form where the coefficients are factored polynomials in z, w. It is a powerful tool for discovering structure in large expressions. Since a special form of this is used often we define

```
simp:=proc(expr,var)
local n,d:
  n:=numer(expr);
  d:=denom(expr);
  collect(n,var,factor)/collect(d,var,factor);
end:
```

With the above f

```
simp(f,y)
```

returns a polynomial in y where the coefficients are factored polynomials in x, z, w.

Here follows the MAPLE session

```
> a:=c^2:

> vdW:=p*V^3-(p*b+RT)*V^2+a*V-a*b; # van der Waals equation

                          3               2    2      2
                      p V  - (p b + RT) V  + c  V - c  b


> D(p):=0:D(RT):=0:D(c1):=0:D(c2):=0:D(b1):=0:D(b2):=0:
> D(c):=al:    # alpha=c2-c1
> D(b):=be:    # beta=b2-b1
> D(al):=0:D(be):=0:
> D(x):=1:

# Computing diff(V,x) from vdW=0:

> Vx1:=solve(D(vdW),D(V)):

# Get rid of p:
```

```
> S3:=D(V)=Vx,p=RT/(V-b)-a/V^2:
> Vx2:=normal(subs(S3,Vx1)):
> Vx:=simp(Vx2,RT);
                                3                    2
                        - be V  RT + 2 al c (V - b)  V
                        -------------------------------
                                  3      2        2
                             - RT V  + 2 c  (V - b)

# The partial derivatives of w are computed

> w:=RT*log(V-b)+a/V-p*V;   # omega
                                          2
                                         c
                          RT ln(V - b) + ---- - p V
                                          V

> wx1:=normal(RT*D(V-b)/(V-b)+D(a/V-p*V)):
> wx2:=subs(S3,wx1): # simple expression
> wx:=simp(wx2,RT);

       7              2      4           2      2
     (V  be (V - b) RT  - 2 V  c (V - b)  (al V  + c be V - c be b) RT

              3  3           4    /           2  7               4  4  2
          + 4 V  c  al (V - b) )  /  (- (V - b)  V  RT + 2 (V - b)  V  c )
                                 /

> wxx1:=normal(D(wx)):
> wxx2:=subs(S3,wxx1):
> wxx:=simp(wxx2,RT);

                           2        2                2
                          V  (V - b)  (- al V + c be)  RT
                  - 2 -----------------------------------
                            5        2         2  2       4
                       - V  (V - b)  RT + 2 V  c  (V - b)

> wxxx1:=normal(D(wxx)):
> wxxx2:=subs(S3,wxxx1):
> wxxx:=simp(wxxx2,RT);

    5                   2   3            3        2                2   2
(6 V  be (- al V + c be)  RT  - 12 c al V  (V - b)  (- al V + c be)  RT

         3        3                3
   - 16 c  (V - b)  (- al V + c be)  RT)

     /     9   3      2        2  6   2       4        4  3         6        6
    /  (- V  RT  + 6 c  (V - b)  V  RT  - 12 c  (V - b)  V  RT + 8 c  (V - b) )
   /

# Computation of the spinodal equation: elimination of p from vdW=0,mu1x=0.
```

```
> mu1x:=normal(x*wxx-RT/(1-x)):   # cf. (8)
> form1:=numer(mu1x):  # mu1x=0 equivalent to form1=0, cf. (10)
> form2:=quo(form1,RT,RT):  # get rid of factor RT

# form2=0 is already the equation of the spinodal.
# Van Laar puts it in the form RT=.... and he uses th=al*V-be*c.

> form3:=solve(form2,RT):
> form4:=subs(al=(th+be*c)/V,form3):
> f:=simp(form4,th):  # RT=f is van Laar's expression for the spinodal
> spin:=RT-f; # cf. formula (13)
                                        2      2        2
                          - 2 x (- 1 + x) th  + 2 c  (V - b)
                    RT - ------------------------------------
                                         3
                                        V

# Computation of the equation of the critical curve: elimination
# of p,T from vdW=0,mu1x=0,mu1xx=0.

> eq1:=(1-2*x)*wxx+x*(1-x)*wxxx: # eq1=0 equiv mu1xx=0, cf. (11')
# eq1=diff(x*(1-x)*wxx,x)
> eq2:=numer(eq1):
# eq3:=simp(op(6,eq2),RT):

> eq4:=subs(RT=f,eq3): # elimination of T
> eq5:=numer(eq4):
> eq6:=factor(eq5):
> eq7:=op(5,eq6):  # only significant factor
> eq8:=simp(eq7,th):

# eq8=0 is the equation of the critical curve. Van Laar replaces
# al*V-c*be (= al*(V-b)+b1*c2-b2*c1) by th.

> S4:={-al*V+c*be=-th,al*V-c*be=th,al*V-2*cb=th-c*be,-al*V+2*c*be=-th+c*be}:
> eq9:=subs(S5,eq8):
> eq10:=simp(eq9,th):  # factor out th
> eq11:=quo(eq10,th,th):
> eq12:=simp(eq11,th): # critical curve almost in van Laar form

> crit:=x*(1-x)*th^3*((1-2*x)*V-3*x*(1-x)*be)
+c*(V-b)^2*(3*x*(1-x)*th*(th-be*c)+c^2*(V-b)*(V-3*b)):

# crit is the critical curve in the van Laar form (15)

> check:=expand(eq12-crit);
                                  check := 0
```

3.2 A singular critical curve

We continue the above MAPLE session, omitting some details. Putting $b_2 = b_1$ and substituting $\beta = 0$, $c = c_1(1-x) + c_2 x$ in *crit*, one obtains *crit*4

```
> crit4;
```

```
           3                                                          2
x (1 - x) th  (1 - 2 x) V + (c1 (1 - x) + c2 x) (V - b1 (1 - x) - b1 x)  (

                 2
    3 x (1 - x) th

                         2
     + (c1 (1 - x) + c2 x)  (V - b1 (1 - x) - b1 x) (V - 3 b1 (1 - x) - 3 b1 x)

    )

# crit4 is homogeneous of degree 4 in v,b1, homogeneous of degree 3 in c1,c2
# Hence put

> crit5:=factor(subs(V=s*b1,c2=t*c1,crit4)):

# crit5 is a viewed as a polynomial in x and s with coefficients
# depending on b1,c1,t

> crit6:=op(2,crit5): # the relevant factor
> critx:=diff(crit6,x):
> crits:=diff(crit6,s):

# Elimination of s:

> res1:=resultant(crit6,critx,s): # contains 385 terms
> res2:=resultant(crit6,crits,s): # contains 930 terms

# Nevertheless quite quick factorization:

> res1f:=factor(res1):
> res2f:=factor(res2):

# The only meaningful factor of resf1 is r1, a polynomial in x and t
# alone containing 49 terms. Similarly, the only meaningful factor
# of res2f is r2, a polynomial in x and t alone having 20 terms.

> r1;
        7                   2       5      7  7       5  5       6
- 1 + 4 x  - 3 t + 10 x - 39 x  + 66 x  + 4 t  x  - 72 t  x  - 25 x

            8  2         6  3       4        6  7         5           2
     - 216 x  t  - 1717 x  t  - 95 x  - 200 t  x  + 2508 x  t - 339 t x

             4             3        5  6         7  2         3  4         2  4
     - 2165 x  t + 1128 t x  + 369 t  x  + 1224 x  t  - 1306 t  x  - 2646 t  x

            8  5         6  2        6  4       3        8  6        5  7
     + 216 x  t  - 2889 x  t  - 159 x  t  + 80 x  + 108 x  t  - 504 t  x

            4  5         5  2        8           6          7  3        8  3
     + 426 t  x  + 3654 x  t  - 108 x  t - 1737 x  t + 700 x  t  - 108 x  t

             2      6  7       5  6       2  3       3  4      5  4       8  4
```

```
     + 18 x t  - 3 x  t  - 18 x  t  - 45 x  t  + 48 x  t  - 9 t  x  + 108 x  t

           7          3  5       4  4       7  4       6  6        3  2
     + 664 x  t + 2076 t  x  - 259 t  x  - 164 x  t  + 113 t  x  + 1080 x  t

                    2  2       3  3
     + 52 t x - 225 t  x  + 400 t  x

> r2;
 5  3       4  4       3  4      2  4       3  4       3  3       2  3
t  x  + 27 t  x  - 35 x  t  + 8 x  t  - 81 t  x  + 142 t  x  - 77 x  t

             3       2  4       3  2       2  2        2       4          3
     + 16 x t  + 81 t  x  - 182 x  t  + 137 t  x  - 36 x t  - 27 x  t + 73 t x

             2            3     2
     - 65 t x  + 19 t x + x  - 3 x  + 3 x - 1

# Final elimination of t

> res:=resultant(r1,r2,t)  # yields polynomial in x alone with 44 terms
> resf:=factor(res):
> xeq:=8*x^4-16*x^3+4*x^2+4*x-1: # only relevant factor
> xeq;
                           4      3     2
                        8 x  - 16 x  + 4 x  + 4 x - 1

# Some thought is needed to see that the other factors of resf are
# irrelevant.

> sol:=[solve(xeq)];
                       1/2        1/2  1/2            1/2        1/2  1/2
    sol := [1/2 + 1/4 2    + 1/4 3    2   , 1/2 + 1/4 2    - 1/4 3    2   ,

                    1/2        1/2  1/2            1/2        1/2  1/2
        1/2 - 1/4 2    + 1/4 3    2   , 1/2 - 1/4 2    - 1/4 3    2   ]

> solfloat:=map(evalf,sol);
       solfloat := [1.465925826, .2411809547, .7588190453, -.4659258263]

# Because of 0 <= x <= 1, only x=0.2411809547,x=0.7588190453 are possible.
# Note that 0.2411809547+0.7588190453=1, reflecting the symmetry of x and
# 1-x. Note also that op(2,sol) is the expression (17) found by van Laar.
```

References

[1] U.K. Deiters, *Phänomenologie und Berechnung von Fluid-Phasengleichgewichten,* Ruhr-Universität, Bochum (1992)

[2] E.P. van Emmerik, *J.J. van Laar (1860-1938), A mathematical chemist*, Ph.D. thesis, Delft (1991)

[3] M. de Haas, *Thermodynamika* (Dutch) P. Noordhoff N.V. (1948)

[4] P.H. van Konynenburg, R.L. Scott, *Critical lines aand the phase equilibria in binary van der Waals mixtures,* Phil. Trans. Royal Soc. London Vol 298 (1980), 495-540

[5] D.J. Korteweg, *La théorie générale des plis*, Archives Néerlandaises, Vol. (I) T. XXIV (1891), 295-368

[6] D.J. Korteweg, *Sur les points de plissement et les plis correspondants dans le voisinage des bords de la surface ψ de van der Waals*, Archives Néerlandaises, Vol. (II) T. VIII (1903) 235-258

[7] R. Kubo, *Thermodynamics,* North-Holland Publishing Company (1968)

[8] J.J. van Laar, *An exact expression for the course of the spinodal curves and of their plait-points for all temperatures, in the case of mixtures of normal substances,* Proc. Kon. Acad. Amsterdam, Vol. VII March 25, (1905), 646-657

[9] J.J. van Laar, *On the shape of the plaitpoint curve for mixtures of normal substances,* Proc. Kon. Acad. Amsterdam, Vol. VIII (1905), 33-48

[10] J.M.H. Levelt Sengers, *Liquidons and gasons; Controversies about the continuity of states,* Physica Vol. 98A (1979), 363-402

[11] A. van Pelt, *Critical phenomena in binary fluid mixtures,* Ph.D. thesis, Technical University Delft (1992)